
CharonAbsinthe is an extension package for Charon to use it with Absinthe.

Documentation
Documentation (including this readme with module links resolved) can be found at https://hexdocs.pm/charon_absinthe.
How to use
Installation
The package can be installed by adding charon_absinthe to your list of dependencies in mix.exs:
def deps do
  [
    {:charon_absinthe, "~> 0.1.3"}
  ]
end
Setup the parent package
Follow the Charon readme until you've created token pipelines.
Create an error handler
defmodule MyAppWeb.Absinthe do
  def authentication_error(resolution, auth_error_msg) do
    message = "request could not be authenticated: \#{auth_error_msg}"
    extensions = %{error: "authentication_failure", reason: auth_error_msg}
    error = %{message: message, extensions: extensions}
    Absinthe.Resolution.put_result(resolution, {:error, error})
  end
end
Configuration
Additional config is required, see CharonAbsinthe.Config.
Configure router
Add a pipeline with the CharonAbsinthe.HydrateContextPlug, route the graphql endpoint through it,
and register the CharonAbsinthe.send_context_cookies/2 callback as an Absinthe.Plug :before_send hook.
defmodule MyAppWeb.Router do
  use MyAppWeb, :router

  @config Application.compile_env(:my_app, :charon) |> Charon.Config.from_enum()

  pipeline :api do
    plug :accepts, ["json"]
  end

  pipeline :charon_auth do
    plug CharonAbsinthe.HydrateContextPlug, @config
  end

  scope "/api" do
    pipe_through :charon_auth

    forward "/graphql", Absinthe.Plug,
      schema: MyAppWeb.Absinthe.Schema,
      before_send: {CharonAbsinthe, :send_context_cookies}
  end
end
Protect your schema
Normal "auth-required" fields must be protected with CharonAbsinthe.ReqAuthMiddleware.
The refresh mutation must be protected with CharonAbsinthe.ReqRefreshAuthMiddleware.
Mutations that alter the session (login, logout, refresh, logout-all, logout-other...) must be followed by CharonAbsinthe.PostSessionChangeMiddleware.
defmodule MyAppWeb.Absinthe.SessionTypes do
  use Absinthe.Schema.Notation
  alias MyAppWeb.Absinthe.SessionResolver
  alias CharonAbsinthe.{ReqAuthMiddleware, ReqRefreshAuthMiddleware, PostSessionChangeMiddleware}

  @config Application.compile_env(:my_app, :charon) |> Charon.Config.from_enum()

  object :session_mutations do
    field :login, type: :login_payload do
      arg :email, non_null(:string)
      arg :password, non_null(:string)
      arg :token_signature_transport, non_null(:string)
      resolve &SessionResolver.login/3
      middleware PostSessionChangeMiddleware
    end

    field :logout, type: :logout_payload do
      middleware ReqAuthMiddleware, @config
      resolve &SessionResolver.logout/3
      middleware PostSessionChangeMiddleware
    end

    field :refresh, type: :refresh_payload do
      middleware ReqRefreshAuthMiddleware, @config
      resolve &SessionResolver.refresh/3
      middleware PostSessionChangeMiddleware
    end
  end
end
Create a session resolver
Error handling is omitted.
defmodule MyAppWeb.Absinthe.SessionResolver do
  alias Charon.{Utils, SessionPlugs}
  alias MyApp.{User, Users}

  @config Application.compile_env(:my_app, :charon) |> Charon.Config.from_enum()

  def login(
        _parent,
        _args = %{token_signature_transport: transport, email: email, password: password},
        _resolution = %{context: %{charon_conn: conn}}
      ) do
    with {:ok, user} <- Users.get_by(email: email) |> Users.verify_password(password) do
      conn
      |> Utils.set_token_signature_transport(transport)
      |> Utils.set_user_id(user.id)
      |> SessionPlugs.upsert_session(@config)
      |> token_response()
    end
  end

  def logout(_parent, _args, _resolution = %{context: %{charon_conn: conn}}) do
    conn
    |> SessionPlugs.delete_session(@config)
    |> then(fn conn -> {:ok, %{resp_cookies: conn.resp_cookies}} end)
  end

  def refresh(
        _parent,
        _args,
        _resolution = %{context: %{charon_conn: conn, user_id: user_id}}
      ) do
    with %User{status: "active"} <- Users.get_by(id: user_id) do
      conn |> SessionPlugs.upsert_session(@config) |> token_response()
    end
  end

  ###########
  # Private #
  ###########

  defp token_response(conn) do
    tokens = conn |> Utils.get_tokens() |> Map.from_struct()
    session = conn |> Utils.get_session() |> Map.from_struct()
    {:ok, %{resp_cookies: conn.resp_cookies, tokens: tokens, session: session}}
  end
end

Absinthe integration for Charon.

        send_context_cookies(conn, blueprint)

      


        Absinthe helper to send any response cookies present in the context.
To be used as a before_send hook for Absinthe.Plug.
Response cookies will be set by CharonAbsinthe.PostSessionChangeMiddleware.

    send_context_cookies(conn, blueprint)

          @spec send_context_cookies(Plug.Conn.t(), Absinthe.Blueprint.t()) :: Plug.Conn.t()


      


Absinthe helper to send any response cookies present in the context.
To be used as a before_send hook for Absinthe.Plug.
Response cookies will be set by CharonAbsinthe.PostSessionChangeMiddleware.

Config module for CharonAbsinthe.
Charon.Config.from_enum(
  ...,
  optional_modules: %{
    CharonAbsinthe => %{
      access_token_pipeline: MyApp.AccessTokenPipeline,
      refresh_token_pipeline: MyApp.RefreshTokenPipeline,
      auth_error_handler: &MyApp.Absinthe.auth_error_handler/2
    }
  }
)
The following options are supported:
	:access_token_pipeline (required). A Plug that validates an access token (see readme).
	:refresh_token_pipeline (required). A Plug that validates a refresh token (see readme).
	:auth_error_handler (required). A function that takes an %Absinthe.Resolution{} struct and an error message, and returns the resolution struct.

        from_enum(enum)

      


        Build config struct from enumerable (useful for passing in application environment).
Raises for missing mandatory keys and sets defaults for optional keys.

      Link to this type

          @type t() :: %CharonAbsinthe.Config{
  access_token_pipeline: module(),
  auth_error_handler:
    (Absinthe.Resolution.t(), String.t() -> Absinthe.Resolution.t()),
  refresh_token_pipeline: module()
}

      Link to this function
    
    from_enum(enum)


      
       
       View Source
     


  


  

      

          @spec from_enum(Enum.t()) :: t()


      


Build config struct from enumerable (useful for passing in application environment).
Raises for missing mandatory keys and sets defaults for optional keys.

Plug to bridge the gap between Plug's conn and Abinthe's context.
Processes the auth token as an access token, and stores the conn in the context as :charon_conn
so that it may be used in Absinthe resolvers for manipulation sessions with Charon.SessionPlugs.
Note that no guarantees are offered about the consistency of the assigns/private fields etc of
this conn struct apart from the Charon-related contents.

Absinthe middleware to update the resolution after a session is created, refreshed or dropped.
Transfers resolution.value.resp_cookies to resolution.

Absinthe middleware to require a valid access token.
Can be used if the context was hydrated by CharonAbsinthe.HydrateContextPlug.

      





  

    
CharonAbsinthe.ReqRefreshAuthMiddleware 
    



      
Absinthe middleware to require a valid refresh token.
Can be used if the context was hydrated by CharonAbsinthe.HydrateContextPlug.

